

Ricerca operativa

Ollari Ischimji Dmitri

4 ottobre 2023

# Indice

<b>1</b>	<b>Modelli</b>	<b>2</b>
1.1	Flusso a costo minimo . . . . .	2
1.1.1	Matrice dei vincoli . . . . .	2
1.1.2	Esempio . . . . .	3
1.2	Flusso massimo . . . . .	4
1.2.1	Modello matematico . . . . .	4
1.2.2	Esempio . . . . .	5

# Capitolo 1

## Modelli

### 1.1 Flusso a costo minimo

Tipico problema di calcolo computazionale che ha le seguenti caratteristiche per i nodi:

- **Nodi sorgente:** nodi che producono il flusso
- **Nodi destinazione:** nodi che consumano il flusso
- **Nodi transito:** nodi che non sono né sorgenti né destinazioni

Per ogni arco  $(i, j) \in A$  sono associati dei costi unitari  $c_{ij}$  per unità di flusso che si spostano da  $i$  a  $j$ , è possibile la presenza di  $d_{ij}$  che indica la capacità massima dell'arco  $(i, j)$ .

Il prodotto *trasmesso* lungo un'arco del grafo è il flusso  $x_{ij}$ , che deve essere minore o uguale alla capacità ma non negativo.

Per calcolare il flusso uscente dal nodo  $i$ :

$$\sum_{j:(i,j) \in A} x_{ij} \quad (1.1)$$

Mentre il flusso uscente dal nodo  $i$ :

$$\sum_{j:(j,i) \in A} x_{ji} \quad (1.2)$$

I vincoli del problema derivanti dalla conservazione del flusso sono:

$$\sum_{j:(i,j) \in A} x_{ij} - \sum_{j:(j,i) \in A} x_{ji} = b_i \quad \forall i \in V \quad (1.3)$$

Il modello matematico completo è:

$$\min \sum_{(i,j) \in A} c_{ij} x_{ij} \quad (1.4)$$

$$\sum_{j:(i,j) \in A} x_{ij} - \sum_{j:(j,i) \in A} x_{ji} = b_i \quad \forall i \in V \quad (1.5)$$

$$0 \leq x_{ij} \leq d_{ij} \quad \text{interi} \quad \forall (i, j) \in A \quad (1.6)$$

#### 1.1.1 Matrice dei vincoli

La matrice che ha tante **righe** quanti sono i **vincoli** e tante **colonne** quante sono le **variabili**.

La matrice dei vincoli di uguaglianza per i problemi di flusso a costo minimo coincide con la matrice di incidenza **nodo-arco** della rete.

	$x_{12}$	$x_{13}$	$x_{15}$	$x_{23}$	$x_{42}$	$x_{34}$	$x_{53}$	$x_{45}$
1	1	1	1	0	0	0	0	0
2	-1	0	0	1	-1	0	0	0
3	0	-1	0	-1	0	1	-1	0
4	0	0	0	0	0	-1	0	1
5	0	0	-1	0	0	0	0	-1

### 1.1.2 Esempio

Rete  $G = (V, A)$  con  $V = \{1, 2, 3, 4, 5\}$  e

$$A = \{(1, 2); (1, 3); (1, 5); (2, 3); (3, 4); (4, 2); (4, 5); (5, 3)\}$$

$(i, j)$	(1, 2)	(1, 3)	(1, 5)	(2, 3)	(3, 4)	(4, 2)	(4, 5)	(5, 3)
$c_{ij}$	5	-2	2	-4	0	6	3	4

$i$	1	2	3	4	5
$b_i$	2	5	1	-4	-4

Figura 1.1: Matrice dei vincoli

Il modello matematico dell'esempio è:

$$\min 5x_{12} - 4x_{23} + 6x_{42} - 2x_{13} + 0x_{34} + 3x_{15} + 4x_{53} + 3x_{45} \quad (1.7)$$

$$x_{12} + x_{13} + x_{15} = 2 \quad (1.8)$$

$$x_{23} - x_{12} - x_{42} = 5 \quad (1.9)$$

$$x_{34} - x_{13} - x_{23} - x_{53} = 1 \quad (1.10)$$

$$x_{42} + x_{45} - x_{34} = -4 \quad (1.11)$$

$$x_{53} + x_{15} - x_{45} = -4 \quad (1.12)$$

$$x_{ij} \geq 0 \quad \forall (i, j) \in A \quad (1.13)$$

La matrice dei vincoli è:

Il modello matematico per la risoluzione scritto in **ampl** è:

---

```

### INSIEMI ###
set NODI ;
set ARCHI within NODI cross NODI ;
### PARAMETRI ###
param b{NODI};
param c{ARCHI};
param d{ARCHI} >= 0, default Infinity ;
### VARIABILI ###
var x{(i, j) in ARCHI} >= 0, <= d[i,j], integer ;
### VINCOLI ###
subject to bilancio{i in NODI} : sum{j in NODI : (i, j) in ARCHI} x[i,j] -
sum{j in NODI :
(j, i) in ARCHI} x[j,i] = b[i] ;
### OBIETTIVO ###
minimize costo_totale : sum{(i, j) in ARCHI} c[i,j]*x[i,j] ;

```

---

I dati del problema sono:

---

```

### INSIEMI ###
set NODI := n1 n2 n3 n4 n5 ;
set ARCHI := (n1,n2) (n1,n3) (n1,n5) (n2,n3) (n3,n4) (n4,n2) (n4,n5)
              (n5,n3) ;
### PARAMETRI ###
param b :=
n1 2
n2 5
n3 1
n4 -4
n5 -4 ;

param c :=
n1 n2 5
n1 n3 -2
n1 n5 2
n2 n3 -4
n3 n4 0
n4 n2 6
n4 n5 3
n5 n3 4 ;

```

---

## 1.2 Flusso massimo

Simile al problema di flusso a costo minimo, ma senza i costi unitari  $c_{ij}$ , quindi si vuole massimizzare il flusso che attraversa la rete.

### 1.2.1 Modello matematico

Le variabili del modello matematico per questo problema sono:

- $x_{ij} \geq 0$ : flusso lungo l'arco  $(i, j)$
- $d_{ij}$ : capacità massima dell'arco  $(i, j)$
- $x_{ij} \leq d_{ij} \quad \forall (i, j) \in A$
- $\sum_{j:(i,j) \in A} x_{ij} = \sum_{j:(j,i) \in A} x_{ji} \quad \forall i \in V \setminus \{S, D\}$  equivalente a dire che il flusso entrante in un nodo è uguale al flusso uscente dal nodo

L'obiettivo del problema è quello di massimizzare la quantità di flusso che esce dal nodo sorgente  $S$ :

$$\sum_{j:(S,j) \in A} x_{Sj} \quad (1.14)$$

Che corrisponde all'aumentare il flusso del nodo destinazione  $D$ :

$$\sum_{j:(j,D) \in A} x_{jD} \quad (1.15)$$

Il modello matematico completo è:

$$\max \sum_{j:(S,j) \in A} x_{Sj} \quad (1.16)$$

$$\sum_{j:(i,j) \in A} x_{ij} = \sum_{j:(j,i) \in A} x_{ji} \quad \forall i \in V \setminus \{S, D\} \quad (1.17)$$

$$0 \leq x_{ij} \leq d_{ij} \quad \forall (i, j) \in A \quad (1.18)$$

**OSS:** La matrice dei vincoli di uguaglianza del problema di flusso massimo coincide con la matrice di incidenza node-arco della rete senza i nodi sorgente e destinazione.

## 1.2.2 Esempio

La rete ha le seguenti caratteristiche:

Arco	(S, 1)	(S, 2)	(1, 3)	(1, 4)	(2, 3)	(2, 1)	(3, D)	(4, D)
Capacità	3	2	1	4	1	1	1	7

Il modello matematico per l'esempio è:

$$\max x_{S1} + x_{S2} \quad (1.19)$$

$$x_{13} + x_{14} - x_{S1} = 0 \quad (1.20)$$

$$x_{23} + x_{24} - x_{S2} = 0 \quad (1.21)$$

$$x_{3D} - x_{13} - x_{23} = 0 \quad (1.22)$$

$$x_{4D} - x_{14} - x_{24} = 0 \quad (1.23)$$

$$0 \leq x_{S1} \leq 3 \quad (1.24)$$

$$0 \leq x_{S2} \leq 2 \quad (1.25)$$

$$0 \leq x_{13} \leq 1 \quad (1.26)$$

$$0 \leq x_{14} \leq 4 \quad (1.27)$$

$$0 \leq x_{23} \leq 1 \quad (1.28)$$

$$0 \leq x_{24} \leq 1 \quad (1.29)$$

$$0 \leq x_{3D} \leq 1 \quad (1.30)$$

$$0 \leq x_{4D} \leq 7 \quad (1.31)$$

$$x_{ij} \geq 0 \quad \text{interi} \quad \forall (i, j) \in A \quad (1.32)$$

La matrice dei vincoli che ne consegue è:

	$x_{S1}$	$x_{S2}$	$x_{13}$	$x_{14}$	$x_{23}$	$x_{24}$	$x_{3D}$	$x_{4D}$
1	-1	0	1	1	0	0	0	0
2	0	-1	0	0	1	1	0	0
3	0	0	-1	0	-1	0	1	0
4	0	0	0	-1	0	-1	0	1

Il modello matematico per la risoluzione scritto in **ampl** è:

---

```

### INSIEMI ###
set NODI ;
set ARCHI within NODI cross NODI ;
### PARAMETRI ###
param Sorgente symbolic in {NODI};
param Destinazione symbolic in {NODI} , != Sorgente ;
param d{ARCHI} >= 0, default Infinity ;
### VARIABILI ###
var x{(i, j) in ARCHI} >= 0, <= d[i,j], integer ;
### VINCOLI ###
subject to equilibrio{i in NODI diff {Sorgente, Destinazione}} : sum{j
    in NODI : (i, j) in

```

```
    ARCHI} x[i,j] - sum{j in NODI : (j, i) in ARCHI} x[j,i] = 0 ;  
### OBIETTIVO ###  
    maximize flusso_uscente : sum{j in NODI : (Sorgente,j) in ARCHI}  
        x[Sorgente,j] ;
```

---